1   REM THIS PROGRAM COMPUTES THE TOTAL SALES FOR EACH OF
2   REM FIVE SALESMEN SELLING THREE DIFFERENT GOODS. THE LIST P
3   REM GIVES THE PRICE OF THE THREE GOODS. THE TABLE S GIVES
4   REM THE INDIVIDUAL ITEM SALES OF THE FIVE SALESMEN, WHERE THE
5   REM ROWS STAND FOR THE ITEMS AND THE COLUMNS FOR THE SALESMEN.
6   REM TAKEN FROM THE BASIC MANUAL, P. 37.
10  FOR I = 1 TO 3
20  READ P(I)
30  NEXT I
40  FOR I = 1 TO 3
50  FOR J = 1 TO 5
60  READ S(I, J)
70  NEXT J
80  NEXT I
90  FOR J = 1 TO 5
100 LET S = 0
110 FOR I = 1 TO 3
120 LET S = S + P(I) * S(I, J)
130 NEXT I
140 PRINT "TOTAL SALES FOR SALESMAN " J, "$" S
150 NEXT J
200 DATA 1.25, 4.30, 2.50
210 DATA 40, 20, 37, 29, 42
220 DATA 10, 16, 3, 21, 8
230 DATA 35, 47, 29, 16, 33
300 END
